



**Střední průmyslová škola a Vyšší odborná škola technická Brno, Sokolská 1**  
**Šablona: Inovace a zkvalitnění výuky prostřednictvím ICT**

**Název: AlphaCAM - soustružení**

**Téma: Postprocesory pro soustružení a jejich editace**

**Autor: Horák Jiří**

**Číslo: VY\_32\_INOVACE\_47-19**

**Anotace:** Postprocesory v systému AlphaCAM – soustružení a jejich editace. DUM je určen pro výuku ve 4. ročníku oboru STROJÍRENSTVÍ zaměřením VŠEOBECNÉ předmět CAD/CAM systémy a zaměřením POČÍTAČOVÉ ŘÍZENÍ NC A CNC STROJŮ předmět Programování CNC strojů.  
Vytvořen: 28.5.2013

Postprocesor je softwarový převodník CL dat z CAD/CAM systému do datového jazyka konkrétního obráběcího stroje. Kvalitní postprocesor v sobě obsahuje veškeré informace o vlastnostech daného stroje, tak aby bylo optimálně a efektivně využito všech jeho funkcí v souladu s CAD/CAM systémem.

Univerzální postprocesor neexistuje, bohužel je potřeba jej naprogramovat pro každý stroj zvlášť. Cesta k dokonalému postprocesoru se stává obtížnější i kvůli faktu, že pravděpodobně neexistují dvě identické konfigurace obráběcího CNC stroje. Počet možných kombinací je velmi vysoký. Když toto uvážíme, není překvapující vysoká možnost nezdaru při neodborném programování.

V podstatě má uživatel tři způsoby jak postprocesor získat:

1. Objednat si postprocesor na míru u výrobce CAD/CAM systému.
2. Prohledat databázi existujících zákaznických postprocesorů CAM systému svého prodejce a porovnávat, který z nich v jeho podmínkách bude pracovat dostatečně dobře.
3. Vytvořit si vlastní postprocesor sám.

Volba toho kterého způsobu závisí zejména na konkrétních podmínkách používaného CAMu, sortimentu CNC strojů uživatele systému, ale i na schopnostech technologa.




Vytvořit si vlastní postprocesor sám? Zdánlivě se jedná o nejtěžší a komplikovanou volbu, ale ve skutečnosti je to právě naopak. Ty nejlepší postprocesory si vytvářejí sami uživatelé.

V systému AlphaCAM je dodáváno několik postprocesorů pro různé řídicí systémy CNC soustruhů. Jsou uloženy v systémové složce :**\AlphaCAM\LICOMDAT\TPOSTS.ALP** s příponou **\*.atp**.

Otevřít a editovat postprocesory můžeme jednak pomocí AlphaEDIT + RS232 Comms dodávaným se systémem AlphaCAM nebo otevřít v libovolném textovém editoru.

- spustíme AlphaEDIT + RS232 Comms
- v menu Soubor vybereme Otevři Postprocesor – Soustružení

- v systémové složce otevřeme zvolený postprocesor



## **Editace a úprava postprocesoru:**

Základním předpokladem pro úspěšnou úpravu výchozího postprocesoru, je dobrá znalost řídicího systému, pro který se bude postprocesor upravovat.

1. Nejprve zkopírujeme upravovaný postprocesor pod vhodným názvem do pracovní složky.
2. Postprocesor otevřeme pomocí AlphaEDIT + RS232 Comms nebo v libovolném textovém editoru.
3. Postprocesor upravíme a uložíme.
4. Zkopírujeme upravený postprocesor do systémové složky  
:\AlphaCAM\LICOMDAT\TPOSTS.ALP\.
5. Spustíme AlphaCAM – soustružení.
6. Vybereme vhodnou součást a obrobíme. Součást by měla obsahovat všechny druhy operací - soustružení, obrábění otvorů, závitování, zápichy atd.
7. Pomocí upraveného postprocesoru vytvoříme CNC program.
8. Program vyzkoušíme. Nejlépe přímo v řídicím systému stroje.

## **Úpravy postprocesorů:**

### **Proměnné postprocesoru:**

Tyto proměnné jsou dvojího typu - **Rezervované AlphaCAM proměnné** a **Uživatelské proměnné**. Můžeme použít oba typy dle potřeby, ale hodnoty můžeme přiřazovat pouze Uživatelským proměnným. (AlphaCAM přiřazuje hodnoty AlphaCAM proměnným). AlphaCAM používá pro své proměnné nejvýše 3 písmena. Pro Uživatelské proměnné je vhodné použít více písmen, aby nedošlo k problémům.

### **Proměnné pro soustružení:**

NTS = Počet chodů závitu pro soustružení  
TR = Rádus špičky při soustružení.  
FP = Kód pro posuv/min, nebo posuv/otáčka u soustružení.  
CS = Kód pro konstantní otáčky, nebo rychlost u soustružení.  
NX, NZ = Nejbližší hodnota X a Z, platí jen v \$80 (výměna nástroje).  
TOX = Hodnota pro najetí nástroje v X-směru.  
TOZ = Hodnota pro najetí nástroje v Z-směru.  
DDP = Průměr rozvinutí.  
AD = Absolutní průměr.

### **Přepínače a Identifikátory**

#### **2-osy:**

TFB = 1 pro nástroj na přední straně zásobníku, 2 na zadní straně.  
TAB = 1, je-li zásobník nad osou rotace, 2, je-li pod osou rotace.  
YOP = 1, pokud se jedná o operaci v ose Y.  
TNV = Vektor pozice oblouku ostří/špičky nástroje (1 - 8, udává vztah středu oblouku špičky k pozici oblouku ostří/špičky nástroje.  
MOT = Druh Operace Obrábění:

čelní=1 podélné hrubování=2 vybrání=3 dokončování=4  
čelní hrubování=5 podélné zapichování=6 čelní zapichování=7

dráha nástroje ručně=8 závitování=9 vrtání=10  
čelní vybrání=11 závitování v ose=12 hrubování kontury=13

### **C/Y osy:**

MOT = Druh Operace

hrubování/dokončování = 1 vybrání kontury=2

spirální vybrání=4 lineární vybrání=5

vrtání=21 přeruš. tříš.=22 závitování=23 senkování=24

### **Pro strojní cykly:**

K definici profilu použijeme jednu ze tří metod:

1. SN pro podprogramové definování profilu.
2. Číslo řádky pro začátek profilu = LNS, Číslo řádky pro konec profilu = LNE.
3. Kód pro konec cyklu (např. \$301). Řádky profilu budou umístěny bezpodmínečně za řádky předpisu fixního cyklu (např. \$300).

L\_R = 1, je-li nástroj nalevo od profilu, = -1, je-li napravo od profilu

CYC = 1, je-li právě vystupující profil fixní cyklus.

### **Pro 'na zakázku navrhované' strojní cykly lze použít:**

SPD = Začátek průměru profilu, SPZ = Začátek profilu Z.

EPD = Konec průměru profilu, EPZ = Konec profilu Z.

### **Pouze pro HRUBOVÁNÍ (podélné nebo čelní):**

WDC = Šířka/Hloubka řezu, STD = Příklad na průměru, STF = Příklad na čele.

#### **Pouze pro zápichy s rovnými boky (podélné nebo čelní):**

GRW = Šířka zápichu. GRL = Hodnota Z na levé straně zápichu. GRR = Hodnota Z na pravé straně zápichu.

GRB = Spodní průměr zápichu, GRT = Horní průměr zápichu. GTW = Šířka nástroje pro zapichování (nelze použít pro jiné nástroje).

### **Pro všechny typy zápichů:**

WDC = Šířka řezu přes zápich (na průměru nebo čele).

STK = Příklad k ponechání pro dokončování, PKD = Vzdálenost hloubení (stejně jako ZP pro vrtání).

### **Pro vrtání v ose rotace:**

ZB = Dno otvoru (špička vrtáku), ZR = Zadaná počáteční hodnota otvoru v ose Z + 1/10 průměru vrtáku.

ZP = Vzdálenost hloubení, FLR = 1 je-li úplné vyjetí ze zahloubení, 0 pro opak.

### **Pro závitování na průměru:**

THD = Hloubka závitu (celková), THF = První hloubka závitu, THA = Boční úhel závitu

THR = Rozdíl počátečního a koncového poloměru u kuželového závitu.

SPR = Počet prázdných řezů., NOC = Počet řezů

### **Pro vnitřní závitování:**

ZB = Konec závitu, ZR = Zadaná počáteční hodnota v ose Z, THP = Stoupání závitu  
SLT = 1, je-li držení pevné, 0 pro vyrovnávací pouzdro.

DW = Prodleva (pouze je-li SLT = 0).

ROT = Kód pro rotaci - VHR, PHR. RRT je vždy opak aktuálního ROT.

### **VÝPIS \$ ČÍSEL PRO NC KÓD:**

#### **Program - začátek a konec**

\$5 \$STORE, \$RUN a \$RECALL

\$10 Úvodní řádky souboru

\$12 Úvodní řádky hlavního programu.

\$15 Následující řádky hlavního programu.

\$17 Následující řádky souboru.

#### **Rychloposuvy:**

\$20 Řádky pro rychloposuv v XZ.

\$27 Řádky pro rychloposuv v C - ose X Z a/nebo C.

\$28 Řádky pro rychloposuv v ose Y.

#### **Posuvy:**

\$40 Lineární posuv.

\$50 Posuv po kružnici, VHR - smysl (autom. rozdělení <180 stupňů).

\$60 Posuv po kružnici, PHR - smysl (autom. rozdělení <180 stupňů).

\$42 Závitování (jednoprůchodové).

\$47 Posuv v C - ose.

\$48 Posuv v Y - ose.

\$57 Posuv po kružnici (VHR-Smysl) v C - ose.

\$58 Posuv po kružnici (VHR-Smysl) v Y - ose.

\$67 Posuv po kružnici (PHR-Smysl) v C - ose.

\$68 Posuv po kružnici (PHR-Smysl) v Y - ose.

#### **Výměna nástroje:**

\$70 Existující nástroj odlož. (Číslo nástroje bylo [T])

\$71 Kód pro konstantní posuv na povrchu, uložen v proměnné [CS].

\$72 Kód pro konstantní otáčky vřetena, uložen v proměnné [CS].

\$73 Kód pro posuv/min, uložen v FP.

\$74 Kód pro posuv/otáčku, uložen v FP.

\$75 Kód pro směr otáček VHR - se do APS v Proměnné RT načte.

\$76 Kód pro směr otáček PHR - se do APS v Proměnné RT načte.

\$80 Zvol nový nástroj. Číslo Nástroje [T], Otáčky [S].

\$84 Volby nástroje pro osu C.

\$86 Změna programovacího bodu nástroje.

\$87 Vložení bodu synchronizace (4 - osy soustružení).

#### **Pracovní roviny:**

\$88 Zvol Pracovní Rovinu.

89 Může být použito při výměně Pracovní Roviny a Nástroje najednou (Viz \$582).

### **Podprogramy:**

- \$90 VYVOLÁNÍ podprogramu. Číslo podprogramu [SN].
- \$100 ZAČÁTEK podprogramu. Číslo podprogramu [SN].
- \$110 KONEC podprogramu.
- \$120 Posunutí nulového bodu. (posunutí o [OX] a [OY]).
- \$130 Posunutí nulového bodu ZRUŠ.

### **Korekce nástroje:**

- \$140 Korekce stroje ZRUŠ.
- \$141 Korekce stroje VLEVO.
- \$142 Korekce stroje VPRAVO.
- \$145 Procentuální zvětšení přechodových vnitřních poloměrů.
- \$146 Přechod. vnitřní poloměry na ostrých vnitřních rozích (1=Ano, 0=Ne)

### **Kódy chlazení:**

Kódy pod \$150 - \$153 jsou AlphaCAM přiřazeny do CLT na začátku každé dráhy.

- \$150 Kód pro ŽÁDNÉ chlazení.
- \$151 Kód pro NORMÁLNÍ chlazení.
- \$152 Kód pro TLAKOVÉ chlazení.
- \$153 Kód pro chlazení V NÁSTROJI.

### **Soustružení - strojní cykly:**

- \$300 Cykly pro podélné hrubování.
- \$301 Podélné hrubování konce profilu(prázdné při použití SN, nebo LNS/LNE).
- \$305 Cyklus pro čelní hrubování.
- \$306 Čelní hrubování konce profilu (prázdné při použití SN, nebo LNS/LNE).
- \$310 Cyklus pro vrtání v ose rotace vřetene.
- \$311 Cyklus pro vnitřní závitování v ose rotace.
- \$320 Cyklus pro zapichování na průměru.
- \$325 Cyklus pro zapichování na průměru s profilem.
- \$326 Zapichování na průměru konce profilu. (prázdné při použití SN, nebo LNS/LNE).
- \$330 Cyklus pro zapichování na čele.
- \$335 Cyklus pro zapichování na čele s profilem.
- \$336 Zapichování na čele konce profilu. (prázdné při použití SN, nebo LNS/LNE).
- \$340 Cykly pro řezání závitů.

### **FORMÁTY OBECNĚ:**

Zápisy se oddělují mezerou. Více znaků je v pořádku, např. X VX VY.

- \$500 Modální ZÁPIS TEXT (v následujících řádcích se neopakují).
- \$502 Modální ABSOLUTNÍ HODNOTA (u stejných hodnot se neopakuje).
- \$504 Modální INKREMENTÁLNÍ HODNOTY (u stejných hodnot se neopakuje).
- \$510 X Y Z I J K Hodnoty musí mít znaménka +/- (1 = Ano, 0 = Ne).
- \$515 Desetiný oddělovač <.> nebo <,> (1 = tečka <.>, 0 = čárka <,>).
- \$520 Podprogram na konec hlavního programu (1 = Ano, 0 = Začátek).
- \$525 Omezení kružnic kvadranty (1 = Ano, 0 = až 180 Stupňů).
- \$530 Oblouky se generují jako lineární úseky (1 = Ano, 0 = Ne).
- \$531 Oblouky jako přímky pro poloměry větší (0 => \$531 se ignoruje).
- \$532 Odchylka sečny pro \$530 = 1.
- \$540 Výstup seznamu operací, START a Komentář potlačit (1=Ano, 0=Ne).

\$533 Výstup oblouku nebo přímky u C-osy jako přímkové segmenty (1=Ano,0=Ne).

\$550 Výstup jako průměr nebo rádius (1 = průměr, 0 = rádius).

\$551 Ignoruj podřezávání v hrubovacím fixním cyklu (1 = Ignoruj, 0 = Výstupem kompletní profil).

\$555 Korekce na řeznou hranu nebo střed rádiusu (1 = řezná hrana, 0 = střed rádiusu).

\$590 Výměník nástrojů pod osou: x hodnoty průměru nebo poloměru jsou kladné (1= Ano, 0=Ne).

Jestliže nastaveno na ano, oblouky ve směru hod. ruč. použijí \$60 (G3), v protisměru hod. ruč. \$50 (G2).

### **FORMÁTY ČÍSEL:**

\$700 Formát čísla pro číslo podprogramu.

\$701 Počáteční číslice.

\$702 Koncové číslice / místa po čárce.

\$705 Startovací číslo pro podprogram.

\$710 Formát čísla pro NC-čísla.

\$711 Počáteční číslice.

\$712 Koncové číslice / místa po čárce.

\$715 Startovací číslo pro NC-číslování.

\$716 Krok pro NC-číslování.

\$720 Formát čísel X Y Z.

\$721 Počáteční číslice.

\$722 Koncové číslice / místa po čárce.

\$730 Formát čísla Střed Oblouku / Poloměr.

\$731 Počáteční číslice.

\$732 Koncové číslice / místa po čárce.

\$740 Formát čísla Otáčky.

\$741 Počáteční číslice.

\$742 Koncové číslice / místa po čárce.

\$743 Maximální otáčky.

\$744 Pevné Otáčky (např. 100, 200, 1000, 2000 ... 0 = zadané pevné hodnoty).

*Poznámka: Pevné otáčky mohou být více než na jednom řádku. Maximum je 80 znaků na řádek. AlphaCAM vybere nejnižší blízkou hodnotu k vypočtené hodnotě. Níže uvedený \$745 se užívá pouze tehdy, je-li \$744 nastaven na 0.*

\$745 Okruh Otáček v násled. stupních: (.1 nebo 1 nebo 10..0 = žádný okruh).

\$750 Formát čísla Posuv.

\$751 Počáteční číslice.

\$752 Koncové číslice / místa po čárce.

\$753 Maximální Posuv (mm/min nebo in/min).

\$755 Okruh Posuvu v násled. stupních: (.1 nebo 1 nebo 10..0 = žádný okruh).

\$760 Formát čísla Číslo Nástroje.

\$761 Počáteční číslice.

\$762 Koncové číslice / místo po čárce.

### **RYCHLOPOSUV A ČAS PRO VÝMĚNU NÁSTROJE:**

Je možné v mm/min nebo in/min.

\$900 XY Rychloposuv.  
\$901 Z Rychloposuv.  
\$902 Střední čas pro výměnu nástroje (sekundy).

### **UŽIVATELSKÉ PROMĚNNÉ:**

\$1000 VARIABLE (formát) "Nápověda <default>". " komentář.

### **UŽIVATELSKÝ NC - KÓD:**

Každé \$číslo mezi \$1100 a \$1119 se může používat pro vytvoření uživatelského NC-kódu. První řádek za \$číslo se objeví v prvním dialogovém okně.

Řádky až k prvnímu \$číslo se objeví ve druhém dialogovém okně. Řádky až k druhému \$číslo se vloží do NC-Kódu.

### **UŽIVATELEM DEFINOVANÉ PŘÍKAZY:**

\$1200 - \$1219 mohou být použity každý pro jeden příkaz, např.  
\$1200

První řádek je jméno příkazu, zobrazené v dialogovém rámečku.

Následují řádky makra k vytvoření požadovaného NC kódu.

\$ konec \$1200 Uživatelsky definované příkazy.

### **MATEMATICKÉ FUNKCE [Advanced Systémy]:**

V makru či postprocesoru můžete použít jakýkoliv vhodný matematický výraz.

Platné výrazy mohou obsahovat:

+ - \* / SQR() ABS() MOD( , ) EXP() LOG() LOG10() INT() SIN() COS()  
TAN() ASIN() ACOS() ATAN() ATAN2( , ) COSH() SINH() TANH()

U všech hodnot úhlů se předpokládá, že jsou ve stupních, nikoliv v radiánech.

P \*\* X znamená mocnění hodnoty P hodnotou X (X nemusí být celé číslo, pokud P je kladné).

MIN(X,Y) vrací menší hodnotu, např. pokud X=2 a Y=3, MIN(X,Y)=2.

MAX(X,Y) vrací větší hodnotu, např. pokud X=2 a Y=3, MIN(X,Y)=3.

PI je hodnota pi = 3.14159.... Například můžete uvést OBVODKR = PI \* PRŮMĚR.  
MOD( , ) může někdy poskytnout elegantní řešení problému.

MOD(X,Y) dává jako výsledek zbytek, který zůstal po dělení hodnoty X hodnotou Y, např. MOD(5,2) = 1 MOD(9,2) = 1 MOD(9,3) = 0 MOD(20,3) = 2 Toto můžeme využít při zjišťování, zda je číslo, řekněme X, liché či sudé:

Jestliže MOD(X,2) = 0, pak musí být sudé. Pokud MOD(X,2) = 1, pak musí být liché.

Máme možnost uvést [MOD(N,999) + 1] místo [N] k vytvoření očíslování řádků od 1 do 999, které se znovu nastaví na hodnotu 1 při dosažení tisícího řádku (N musí být nastaveno na počátek od 0 s přírůstkem 1). Pokud je N menší než 999, výsledek dělení je 0 a zbytek je N a bude-li N mezi 999 a 2 x 999, výsledek dělení bude 1 a zbytek bude N - 999 a tak dále.

### **DEKLARACE UŽIVATELSKÝCH PROMĚNNÝCH:**

U všech proměnných musí být deklarována jména a formáty v \$1000. Dle potřeby může být deklarován dotaz i s defaultní hodnotou. Jména proměnných mohou mít nejvíce

20 písmen, přičemž mohou obsahovat i čísla, ale první znak musí být písmeno. Vždy však používejte více než 3 znaky.

Jména proměnných nesmějí obsahovat mezery. K oddělení slov použijte znak ' \_ ', např. X\_NULA. Dotazy mohou mít maximálně 50 znaků. Předdefinované hodnoty musí být v závorkách < >.

Formát pro výstup proměnných (číselných, celočíselných apod.) do NC-kódu se udává v ( ).

#### **Pro textové proměnné :**

(" ") nebo (" ") znamená, že text vystoupí tak jak je zadán.

("8") znamená, že jen prvních 8 znaků vystoupí.

("\_8") znamená, že počáteční mezery vystoupí, aby délka pole dosáhla 8.

("8\_") znamená, že koncové mezery vystoupí, aby délka pole dosáhla 8.

Pro číselné proměnné zadejte ( ) má-li být formát stejný, jako jsou XYZ atd. posuvy, nebo mají-li být hodnoty použity pouze pro výpočty a nevystupují v NC kódu.

Berme v úvahu, že je-li \$510 nastaven na 1 (=výstup znamének +/- před čísly), potom proměnné s formátem ( ) nebudou mít stejný formát jako XYZ čísla, ale budou vystupovat i s + znaménkem u pozitivních hodnot.

Nehledě na výstupní formát, čísla ve výpočtech, nebo přiřazeních jsou používána vždy s maximální přesností a v případě, kdy je třeba je uzpůsobit určitému formátu jsou zaokrouhlována.

#### **Některé příklady:**

\$1000

Poznámka: Uvozovky " uvnitř závorek označují textovou proměnnou. Dva jednoduché apostrofy ' (ne uvozovky) označují poznámku, která nemá být začleněna do NC kódu. Jeden apostrof ' označuje poznámku, která má být začleněna do NC kódu, ale nemá být poslána do stroje.

\$1000

PROGNUM (" ") "Číslo\_programu:" ' Text vystoupí, jak byl zadán.

PROGNAME ("8\_") "Jméno\_programu:" ' Text, doplní se koncovými mezerami.

MY\_NAME (" ") "Programátor: <SCSN>" ' Text vystoupí, jak byl zadán, ale s předdefinovanou hodnotou.

SAFE\_Z ( ) "Vystoupí do NC-kódu ve stejném formátu jako XYZ - hodnoty.

HEAD (6,0,0) "Číslo\_hlavy:" "Vystoupí jako integer.

EIGHT ( ) "Žádný Dotaz, slouží k uložení výpočtu.

OPMSG1 (" ") = "(Vyměň upínku)" " = znamená, že text bude použit, pokud

DWELL\_TIME (" ") = "F2" "mu nebude přiřazena nová hodnota.

*Poznámka: Přiřazení může být provedeno pomocí =, ale výraz nemůže být použit, když je deklarována proměnná, např. MY\_VAR = SAFE\_Z + 2 není povoleno. Použijte \$LET MY\_VAR = SAFE\_Z + 2.*

#### **PŘEDPIS PRO NC KÓD:**

Všechny proměnné a vypočítávané hodnoty na řádku předpisu, který vytváří NC kód musí být v [ ] závorkách. Všechno, co není v hranatých závorkách se bere jako text. Jinak se používají stejná pravidla jako v BASICu, nebo jiných programovacích jazycích a [ ] závorky se nesmí používat ve výrazech, nebo přiřazeních.

Například:

N[N] G01 X[AX]Y[AY] je předpis pro NC kód a musí mít proměnné v [ ].

N[N] G00 Z[ZM + TR/2] je předpis pro NC kód, s [ ] okolo výrazu, ale ne okolo proměnných uvnitř výrazu.

\$LET Z\_UP = ZM + TR/2 je to přiřazení s proměnnými, které nejsou v [ ].

Hodnoty proměnných se berou jako čísla, pokud nejsou v uvozovkách, kdy se berou jako text. Přesto je-li hodnota v uvozovkách a vypadá jako číslo, může být v přiřazeních použito jako číslo.

### **Přiřazení a podmínku:**

Prvním znakem na řádce u těchto příkazů musí být znak \$.

Následující příkazy nemohou být zkracovány. Na rozdíl od AlphaCAM příkazů (např. \$Zoom Všechno) se musí zadávat celé. Je jedno, zda jsou psány velkými, nebo malými písmeny, např. \$if je stejné jako \$IF.

\$LET	Začíná řádek pro přiřazení hodnoty uživatelské proměnné.
\$IF	Začíná řádek s podmínkovým příkazem.
\$ELSE	Nepovinné, umožňuje druhou volbu, jestliže podmínka není pravdivá.
\$ELSEIF	Nepovinné, další podmínku, jestliže předchozí není pravdivá.
\$ENDIF	Ukončuje podmínku. Obvykle následuje NC předpis.

Porovnávat můžete pouze s číslem (ne s textem). Např. jestliže uživatelská proměnná X\_VAL je deklarována jako číslo v \$1000 (viz níže), pak můžete mít:

```
$IF AX > X_VAL
```

```
$LET HLAVA = 2
```

```
$ELSE
```

```
$LET HLAVA = 1
```

```
$ENDIF          ($ENDIF ukončuje $IF řádek)
```

Od výše uvedeného omezení je ale výjimka, je-li proměnná deklarována jako text a hodnota proměnné vypadá jako číslo (text), AlphaCAM ji použije jako číslo a může se provést porovnání jako v horním příkladu.

\$MODAL ON - nastavuje modální kód na ZAPNI (je to defaultní nastavení).

\$MODAL OFF - nastavuje modální kód na VYPNI. Používá se pouze pro \$ číslo, kde se definuje předpis pro NC kód, potom se automaticky přepne na ZAPNI např. u \$50 - posuv po oblouku VHR - Heidenhein potřebuje X a Y pro všechny středy oblouků.

```
$MODAL OFF
```

```
N[N] CC X[AI] Y[AJ]
```

```
$MODAL ON          (Přepne se modální ZAPNI a F se může opakovat)
```

```
N[N] C X[AX] Y[AY] DR-R[TC] F[F] M
```

\$MODAL ON/OFF platí pro všechna adresy uvedené v \$502, ale bez ohledu na to, co je nastaveného v \$502 můžete specifikovat vlastní adresy (každá oddělená čárkou), např.

```
$MODAL OFF X, Y. (Platí jen pro aktuální $ číslo).
```

```
$LL nn - Návěští nn pro GOTO skok.
```

Návěští platí jen v rámci aktuálního \$ číslo předpisu. Např. předpisy \$40 a \$50 a \$60 mohou používat stejná návěští.

\$GOTO nn - Nepodmíněný skok na návěští nn. (Je dovoleno také \$GOTO \$LLnn).

\$IF GOTO nn - Podmíněný skok. (Je dovoleno také \$IF .. \$GOTO \$LLnn).

Skoky mohou být dopředu, nebo dozadu. Jsou povoleny skoky ven z \$IF, \$ELSE, \$ENDIF, ale skoky dovnitř nejsou dovoleny.

Znaménka +, -, \*, /, <, =, > jsou použita v podmínkových příkazech s jejich obvyklým matematickým významem - 'plus', 'minus', atd. Toto umožňuje jednoduché vytváření pružných podmínkových příkazů, např.

Normální NEBO \$IF MC + FF > 0 jestliže MC=1, nebo FF=1, nebo oboje=1.

Vylučovací NEBO \$IF MC 1 FF = 1 jestliže buď MC=1, nebo FF=1, ale ne oba.

ANO \$IF CM + FF = 2 jestliže MC=1 a FF=1.

NEROVNOST \$IF HLAVA < > STARA\_HLAVA.

\$exit - Okamžitě ukončí postprocesor a vrátí do hlavní obrazovky.

Například nemůže-li být při soustružení proveden strojní cyklus, vložte do \$č.pro strojní cyklus.

\$msg Cyklus není dostupný

\$exit

### Ukázka upraveného postprocesoru pr Sinumerik 810T:

\$----- HLA V I C K A -----

Zakaznik : SPSS a VOST Sokolska 1 BRNO

Stroj : EMCO TURN 50/120

ridici system: SINUMERIK 810T

Jednotky : mm

Pocet os :

Sour. system : Absolutni

Contact :

Tel

Fax :

Upravil :

Verze 1 datum: 4.2.2013

Poznamka : Predosove obrabeni

Posledni edit: 4.2.2013

\$-----

Nazev postprocesoru: EMCO TURN 50 (SINUMERIK 810/820T)

\$----- PROGRAM ZACATEK/KONEC -----

\$10 uvodni radek proram

N[MOD(N,9999)+1] (Nazev soucasti: [NAZEV])

\$12 Uvodni radky hlavniho programu

N[MOD(N,9999)+1] G90 G54

N[MOD(N,9999)+1] G58 X0 Z[P\_POSUN]

```

$15 Nasledujici radky hlavniho programu
N[MOD(N,9999)+1] M30
$----- RYCHLOPOSUV -----
$20 Rychloposuv v XZ
$IF TOOL_CHANGE = 0 GOTO 99
$LET AX_MM=AX
$LET AZ_MM=AZ
N[MOD(N,9999)+1] G0 X[AX_MM] Z[AZ_MM]
$ENDIF
$LL 99
$LET TOOL_CHANGE = 1
$----- PRACOVNI POHYB STROJE -----
$40 Linearni interpolace - pohyb po primce
$LET AX_MM=AX
$LET AZ_MM=AZ
N[MOD(N,9999)+1] G1 X[AX_MM] Z[AZ_MM] F[F]
$-----
$42 rezani zavitu - jednopruhodove
$LET AX_MM=AX
$LET AZ_MM=AZ
N[MOD(N,9999)+1] G33 X[AX_MM] Z[AZ_MM] K[F] I[F]
$-----
$50 Kruhova interpolace ve smeru hod. rucicek (APS automaticky omezi uhel 180 jako
maximalni)
$LET AX_MM=AX
$LET AZ_MM=AZ
$LET II_MM=II
$LET IK_MM=IK
N[MOD(N,9999)+1] G2 X[AX_MM] Z[AZ_MM] I[II_MM] K[IK_MM] F[F]
$-----
$60 Kruhova interpolace proti smeru hod. rucicek (APS automaticky omezi uhel 180 jako
maximalni)
$LET AX_MM=AX
$LET AZ_MM=AZ
$LET II_MM=II
$LET IK_MM=IK
N[MOD(N,9999)+1] G3 X[AX_MM] Z[AZ_MM] I[II_MM] K[IK_MM] F[F]
$----- VYMENA NaSTROJE -----
$71 K'd pro konstantni reznou rychlost - vstup do promenne [CS]
96
$72 K'd pro konstantni otacky vretene - vstup do promenne [CS]
97
$73 K'd pro posuv za minutu (F/min) - vstup do promenne [FP]
94
$74 K'd pro posuv na OTACKU (F/ot) - vstup do promenne [FP]
95
$75 K'd pro smer otaceni vretene proti smeru hodinovych rucicek - VPRAVO vstup do
promenne [RT]
M4

```

\$76 Křd pro smer otaceni vretene ve syslu hodinovych rucicek - VLEVO vstup do promenne [RT]

M3

\$-----

\$ Vyber noveho nastroje pro obrabeni. Pouzije se [T] cislo noveho nastroje  
[S] bude zaviset na [CS] a max. prumeru soucasti

\$80

\$LET TOOL\_CHANGE = 1

\$LET POM1 = [CS]

\$LET POM = POM1 + CHLAZENI

\$IF POM=96 "G96 - KONSTANTNI REZNA RYCHLOST

N[MOD(N,9999)+1] G[FP] "G94/G95 - POSUV ZA MINUTU/NA OTACKU

N[MOD(N,9999)+1] G92 S[MS]

N[MOD(N,9999)+1] G[CS] S[S]

N[MOD(N,9999)+1] T[T] D[T] [RT]

\$ENDIF

\$IF POM=98 " G96+M8 - KONSTANTNI REZNA RYCHLOST + CHLAZENI

N[MOD(N,9999)+1] G[FP] "G94/G95 - POSUV ZA MINUTU/NA OTACKU

N[MOD(N,9999)+1] G92 S[MS]

N[MOD(N,9999)+1] G[CS] S[S]

N[MOD(N,9999)+1] T[T] D[T] [RT] ] M8

\$ENDIF

\$IF POM=97 " G97 - KONSTANTNI OTacKY

N[MOD(N,9999)+1] G[FP] "G94/G95 - POSUV ZA MINUTU/NA OTACKU

N[MOD(N,9999)+1] G[CS] S[S]

N[MOD(N,9999)+1] T[T] D[T] [RT]

\$ENDIF

\$IF POM=99 " G97 + M8 KONSTANTNi OTacKY + CHLAZENi

N[MOD(N,9999)+1] G[FP] "G94/G95 - POSUV ZA MINUTU/NA OTACKU

N[MOD(N,9999)+1] G[CS] S[S]

N[MOD(N,9999)+1] T[T] D[T] S[S] M3 M8

\$ENDIF

\$----- KOREKCE -----

\$140 Zruseni koreki

G40

\$141 Korekce nastroje VLEVO

G41

\$142 Korekce nastroje VPRAVO

G42

\$----- OBECHNY FORMAT -----

Separate modal values with spaces. Vice jak jedna adresa je OK eg X VX VY

\$500 Modalni zapis TEXT (Nebude opakovan v nasledujicich radcich napr. G0 G1 G2 G3)

\$502 Modalni ABSOLUTNi HODNOTA (Nebude opakovan v nasledujicich radcich je-li hodnota stejna]

F

\$504 Modalni INCREMENTaLNi HODNOTA (Neprevede se do NC Křdu je-li hodnota = 0)

I K

\$510 X Y Z I J K R hodnoty musi mit znamenko +/- (1 = ANO, 0 = NE)

0

\$515 Desetinna tecka <.> (1 = tecka, 0 = carka <,>)  
 1  
 \$525 Omezeni kruznice kvadranty (1 = ANO, 0 = az 180ř)  
 0  
 \$530 Oblouky se generuji jako linearni useky (1 = ANO, 0 = NE)  
 0  
 \$532 Odchylka secny pro \$530 = 1. Also for C-Axis end milling.  
 .01  
 \$540 Vynechat serizovaci list, START a operacni hlaseni v NC křdu (1 = ANO, 0 = NE)  
 1  
 \$550 Generovat prumer nebo radius v NC křdu (1 = prumer, 0 = radius)  
 1  
 \$555 Korekce na reznou hranu nebo stred radiusu (1 = rezna hrana, 0 = stred radiusu)  
 1  
 \$----- FORMAT CISEL -----  
 \$700 PODPROGRAM ciselny format  
 0  
 \$701 Format qisel pred des. teckou  
 0  
 \$702 Format cisel za des.teckou  
 0  
 \$705 Pocatecni cislo podprogramu  
 1  
 \$-----  
 \$710 Format ciSLOVaNi raDKu  
 4  
 \$711 Format cisel pred des.teckou  
 0  
 \$712 Format cisel za des. teckou  
 0  
 \$715 cislo prvnio radku  
 -1  
 \$716 Krok cislovani radku  
 2  
 \$-----  
 \$720 Format X Z adres  
 2  
 \$721 Format cisel pred des.teckou  
 0  
 \$722 Format cisel za des.teckou  
 4  
 \$-----  
 \$730 Format ARC stred/radius  
 2  
 \$731 Format cisel pred des.teckou  
 0  
 \$732 Format cisel za des.teckou  
 4  
 \$-----  
 \$740 Format OTAcEK VrETENE

4  
\$741 Format cisel pred des.teckou  
0  
\$742 Format cisel za des.teckou  
0  
\$743 Implicitni max. otacky vretene - vstup do promenne [MS]  
3000  
\$-----  
\$750 Format POSUVU STROJE  
2  
\$751 Format cisel pred des.teckou  
0  
\$752 Format cisel za des.teckou  
4  
\$753 MAXIMaLNi POSUV (mm/min)  
750 mm/min  
\$-----  
\$760 Format ciSOVaNi NaSTROJu  
2  
\$761 Format pred des. teckou  
0  
\$762 Format za des. teckout  
0  
\$----- RYCHLOPOSUV A DOBA VYMENY NASTROJE -----  
Uziti m/min nebo in/min je vhodne na tomo miste  
\$900 XZ Rychloposuv  
750 mm/min  
\$902 cas pro vymenu nastroje (sekund)  
20  
\$----- UZIVATELSKEE PROMENNE -----  
Vsechny uzivatelem definovane promenne musi byt deklarovany. max. 20 znaky.  
Delka text promenne max. 50 znaku a implicitni hodnota v <>.  
(" ") Format textove promenne. ( ) Format cisel stejny jako format X Y Z  
Jestlize je pozadovan jiny format musi se zadat (6,0,0) = cislo

\$1000  
OPMSG (" ") "Zadej operacni hlaseni"  
TOOL\_CHANGE ()  
AX\_MM (3,0,3)  
AZ\_MM (3,0,3)  
II\_MM (3,0,3)  
IK\_MM (3,0,3)  
R\_MM (3,0,3)  
P\_POSUN (3,0,3) "Zadej posunutí v ose Z pro G58"  
NAZEV (" ") "Zadej nazev soucasti"  
CHLAZENI (6) "Zapnout chlazení? (NE=0; ANO=2) <2>"  
POM (6)  
POM1 (1)  
\$----- UzIVATELEM DEFINOVANy NC KfD -----  
Prompts up to 30 chars max. Variables up to 20 chars.

```

$1100
Korekce drahy naastroje VLEVO - G41
$-----
N[MOD(N,9999)+1] G41
$-----
$1101
Korekce drahy nastroje VPRAVO - G42
$-----
N[MOD(N,9999)+1] G42
$-----
$1102
Zruseni korekci - G40
$-----
N[MOD(N,9999)+1] G40
$-----
$ ----- Konec postprocesoru-----

```

*Použitá literatura:*

HELP: AlphaCAM Alpha V-7 10.JAN.2007

<http://web.tuke.sk/fvtpo/journal/pdf08/3-str-18-20.pdf>